The Timer

The Timer class is built-in to Flash. Instead of using the frame event to move a ball we will use a timer. The timer can give use more accurate movement.

- Start a new ActionScript 3 file.
- Make the stage 500x300.
- Set the frame rate to 12 frames per second.
- In the ActionScript window type the following:

```
var counter:int=0;
this.addEventListener(Event.ENTER_FRAME,frames);
function frames(e:Event):void {
    this.graphics.beginFill(0x000000);
    this.graphics.drawCircle(counter*5,5,2);
    counter++;
} //frames
```

You will see a dotted black line going across the stage: one dot every frame.

Now, let's do the same thing using a timer. Add this code to the same program. The timer is initialized with the number of milliseconds.

```
var timerRate:int=83;
var dotTimer:Timer=new Timer(timerRate);
dotTimer.addEventListener(TimerEvent.TIMER, drawDot);
dotTimer.start();

function drawDot(e:TimerEvent):void {
      this.graphics.beginFill(0xFF0000);
      this.graphics.drawCircle(e.target.currentCount*5,15,2);
} //drawDot
```

The timer will draw a red dotted line down 15 from the top. The two lines will be drawn at about the same rate because 12 frames per second is about 1 frame every 83 milliseconds. (1000/12).

Things get interesting when we draw the dot based on how much time has elapsed. Add this code to the same program:

```
var dotTimer2:Timer=new Timer(timerRate);
dotTimer2.addEventListener(TimerEvent.TIMER, drawDot2);
var lastTime:Number=getTimer();
dotTimer2.start();

function drawDot2(e:TimerEvent):void {
 var now:int=getTimer();
 var lapsedTime:int=now-lastTime;
 lastTime=now;
```

```
  this.graphics.beginFill(0x00FF00);
  this.graphics.drawCircle(e.target.currentCount*5,
    lapsedTime+20,2);
}//drawDot2
```

Test the movie and move the mouse vigorously when the dotted line is part way across the stage. While the black and red lines are drawn on one vertical line, the green line is more jagged. This is not because the first two methods are better, but because the green line shows that the frame and timer events can vary depending on what other demands are made of the computer processor.



Time Based Animation

We will add a ball to the movie and move it using the timer.

- Create a movie clip called Ball and place two copies on the stage named ball1 and ball2.
- Declare a variable dx:
  `var dx:int=5;`
- Add this line to the frame event:
  `ball1.x+=dx;`
- Add this line to drawDot2:
  `ball2.x+=dx*lapsedTime/100;`

Test the movie.

Now you are ready to experiment!

- Modify the document and try frame rates of 12, 24 and 5.
- Change the value of dx to 1, then try other values.
- Change the timer rate to 400, then try other values.

Notice that ball2 always moves at the same rate, even with different frame rates and timer rates. The only thing that affects the speed of ball2 is `dx`, which is exactly what we want!

Not all processors can keep up with the frame rate you set. If you are trying to match a movie with sound, or want an animation to always appear at the same rate regardless of the processor, timer animation will give you better results than frame animation.